



TDK SPICE Netlist Library の使い方と応用例

アプリケーションセンター 江畑克史

Aug. 10, 2006 AN-NL06B001_ja

はじめに

近年、シミュレータを用いた回路設計が広く行われております。これに対応して、TDKでは電子部品の等価回路をSPICEのネットリスト形式で記述したモデル集「TDK SPICE Netlist Library」をライブラリとして公開しております。このライブラリを用いることにより、実際の電子部品が有する寄生因子を回路シミュレーションに取り入れることができます。

本アプリケーションノートでは、「TDK SPICE Netlist Library」の概要、SPICE シミュレータでの基本的な使い方、および簡単な応用例について述べさせていただきます。

ライブラリの概要

実際の電子部品は高周波領域で顕在化する種々の寄生因子を有するため、回路シミュレーションにおいて電子部品を理想的なものとして扱うと実際の特性に合わないという問題が生じます。そこでTDKでは各種電子部品の等価回路をSPICEのネットリスト形式で表したシミュレーションモデル集「TDK SPICE Netlist Library」(以下、ライブラリ)を作成いたしました。ライブラリには寄生因子を考慮した等価回路モデルが収録されており、これを用いることで実際に即した周波数特性を回路シミュレーションに取り入れることができます。モデルはSPICE ネットリストのサブサーキットと呼ばれる形式で記述されています。SPICEをはじめとする多くの回路シミュレータがこの形式に対応しておりますので、様々な回路シミュレータでライブラリをご使用いただくことができます。

2006年7月現在、ライブラリにはフェライトビーズ(146製品)、3端子フィルタ(70製品)、コモンモードフィルタ(34製品)、パルストランス(6製品)の計256製品のモデルが収録されています。

ライブラリの基本的な使用方法

ライブラリの入手方法

ライブラリはTDKのホームページ(<http://www.tdk.co.jp/tst>)よ

りダウンロードすることができます。ダウンロードファイルはzip形式で圧縮されておりますので、ライブラリのご使用前にzipファイルを解凍して下さい。また、ライブラリをご使用する上での注意点などを記したreadmeファイルもございますので、ライブラリのご使用前に必ずご一読ください。

モデルの内容

ライブラリに収録されているモデルの一例を示しながら、ライブラリの基本的な使用方法についてご説明いたします。モデルを記述したファイルは製品毎に用意されております。ファイル拡張子は「mod」ですが、ファイル自体はテキストファイルですので、テキ

```

*-----
* SPICE Netlist Generated by TDK Corporation
* TDK P/N: ACM2012-900-2P (Common Mode Filter)
* Property: Zc(at 100MHz) = 90 [ohm]
* Model Generated on July 11, 2006
*-----
* External Node Assignments:
*
* 1 ---@@@--- 4
*      ===
* 2 ---@@@--- 3
*
*-----
.SUBCKT ACM2012-900-2P 1 2 3 4
C11 11 12 0.84p
C12 21 22 0.84p
C13 31 32 0.84p
C14 41 42 0.84p
C21 12 42 0.02p
C22 22 32 0.02p
C41 12 22 0.95p
C42 32 42 0.95p
K11 L21 L22 0.999999999
L11 11 12 3.1n
L12 21 22 3.1n
L13 31 32 3.1n
L14 41 42 3.1n
L21 12 42 200n
L22 22 32 200n
R11 11 12 14
R12 21 22 14
R13 31 32 14
R14 41 42 14
R21 12 42 880
R22 22 32 880
R31 1 11 0.065
R32 2 21 0.065
R33 3 31 0.065
R34 4 41 0.065
R41 12 22 1G
R42 32 42 1G
R51 1 4 1G
R52 2 3 1G
.ENDS ACM2012-900-2P
*-----

```

図1 ACM2012-900-2Pのモデル

ストエディタ等で内容の確認や編集が行えます。ここでは、例として差動信号ライン用コモンモードフィルタ「ACM2012-900-2P」について述べます。この製品のモデルは「ACM2012-900-2P.mod」というファイルに記されており、その内容を図 1 に示します。モデルは SPICE ネットリストのサブサーキットと呼ばれる形式で記述されており、サブサーキットとは、特定の回路を一つのパッケージとして予め定義しておき、必要な時に何度でも呼び出して使える機能です。特定の回路を複数箇所に配置したり、回路の一部を様々なサブサーキットに置き換えて特性の変化を調べたりする際に便利な機能です。なお、ライブラリでは、モデルのファイル名およびサブサーキット名には、製品名をそのまま用いております。

次に、モデルの中身について述べます。「*」(アスタリスク)で始まる行はコメント行であり、実際の計算には用いられません。しかし、この部分には製品名、製品カテゴリ、製品の代表特性、サブサーキットの外部ノード割当てなどの情報が記されています。サブサーキット自体の定義は「.SUBCKT」の行から「.ENDS」の行までであり、この間には「ACM2012-900-2P」の等価回路がネットリストという形式で記述されています。この等価回路は実際の電子部品に内在する寄生因子を考慮して作成されていますので、高周波における実際の電気特性を考慮したシミュレーションが可能です。なお、本アプリケーションノートでは等価回路の詳細については割愛させていただきます。

外部ノードの割当て

次に、サブサーキットの外部ノードの割当てについて説明いたします。図 1 では、サブサーキットの定義部分の先頭部に、

```
.SUBCKT ACM2012-900-2P 1 2 3 4
```

と記されてます。これは、そのサブサーキットが「ACM2012-900-2P」という名前前で、1, 2, 3, 4 という番号の4つの外部ノードを有することを示しています。サブサーキットは、これらの外部ノードを通して外部回路と接続されますので、モデルを使用する際には、サブサーキットの外部ノードの割当てを正しく認識し、外部回路と正しく接続することが重要です。この例の場合、外部ノード 1, 2, 3, 4 は、モデル中に絵文字の形式で

```
* 1 ---@@@--- 4
*      ===
* 2 ---@@@--- 3
```

と記された端子に対応しています。従って、このモデルをコモンモードフィルタとして動作させるため

には、ノード 1 と 2 のペアおよびノード 4 と 3 のペアをそれぞれ差動信号の入出力ノードとすれば良いことが判ります。

なお、外部ノードの割当ては製品により異なります。ご使用になるモデル中の絵文字や、ライブラリに含まれる readme ファイルをご参照ください。

モデルの基本的な使用方法

次に、ライブラリを SPICE シミュレータで使用方法を説明いたします。ここでは、最も基本的なテキスト(ネットリスト)ベースのシミュレータで使う場合について記します。なお、SPICE ネットリストの詳細につきましては参考文献[1]、または SPICE シミュレータをご利用であればそのマニュアルなどをご参照ください。

モデルを使用するのに必要な作業は2つあります。一つはモデルが記述されているファイルの内容を、モデルを使用したい回路のネットリスト中にコピー & ペーストすること(インクルード機能があるシミュレータでは、それを使うこともできます)。もう一つはサブサーキットを呼び出すための記述を加えることです。後者について簡単に説明しますと、SPICE ネットリストでは、サブサーキットを呼び出すための記述として名前が「X」で始まるデバイスを使用します。例えば、「ACM2012-900-2P」のサブサーキットを呼び出す場合には、以下のような記述を使います。

```
XCMF 12 22 23 13 ACM2012-900-2P
```

ここでは、例として「XCMF」というデバイス名を用いましたが、「X」で始まる文字列であれば他のデバイス名を使うこともできます。「XCMF」に続く4つの数字は、このサブサーキットが接続される外部回路のノード番号を示します。上の記述では、「ACM2012-900-2P」のサブサーキットの外部ノード 1 は外部回路のノード 12 に、以下順番に 2 は 22 に、3 は 23 に、4 は 13 に、それぞれ接続されます。ここで注意すべき点は、サブサーキットの外部ノード番号と、これと接続される外部回路のノード番号は一致しなくても良いということです。重要なのは番号ではなく、記述さ




図 2 サブサーキットの内部構造と外部回路との接続




図3 差動伝送回路

れる順序です。この例の場合、サブキットの内部構造と外部回路との接続の関係は、図2のようになります。すなわち、外部回路のノード12と22のペア、および13と23のペアが、それぞれ差動信号の入出力ノードとして働くことになります。

以上の様に、モデルを使用する際には絵文字等を参考にして外部ノードの割当てを認識し、外部回路と正しく接続することが重要です。

● ライブラリの簡単な応用例

次に、簡単な例を用いながらライブラリの応用例について述べます。ここでは、図3に示すようなUSB2.0の差動伝送回路のシミュレーションを考えます。この回路の例では、簡単のため逆相で励振された2本のシングルエンドラインを用いて差動回路を模擬します。なお、図中の斜体数字はノード番号を示しています。V1とV2はパルス電源、RIN1とRIN2は信号源の出力インピーダンス、T1とT2は無損失伝送線路、ROUT1とROUT2は負荷抵抗、XCMFはコモンモードフィルタです。

V1とV2は互いに逆位相のパルス電源であり、RIN1、RIN2と共に差動信号源を構成します。シミュレーション上のパルス電源の出力インピーダンスはゼロですが、現実の信号源は有限の出力




図4 シミュレーションで用いる信号源の出力波形

```

***<< simulation of USB2.0 circuit >>*****
*---< main circuit >---*
V1 11 0 PULSE 0.0 0.8 0.0N 0.4N 0.4N 1.68N 4.16N
V2 21 0 PULSE 0.8 0.0 0.1N 0.4N 0.4N 1.68N 4.16N
RIN1 11 12 45
RIN2 21 22 45
XCMF 12 22 23 13 ACM2012-900-2P
T1 13 0 14 0 Z0=45 TD=2N
T2 23 0 24 0 Z0=45 TD=2N
ROUT1 14 0 45
ROUT2 24 0 45
.TRAN 0.05N 10N
.PRINT TRAN V(14) V(24)
*
*---< subckt >---*
*
* SPICE Netlist Generated by TDK Corporation
* TDK P/N: ACM2012-900-2P (Common Mode Filter)
* Property: Zc(at 100MHz) = 90 [ohm]
* Model Generated on July 11, 2006
*
* External Node Assignments:
*
* 1 ---@@@--- 4
*      ===
* 2 ---@@@--- 3
*
*-----
.SUBCKT ACM2012-900-2P 1 2 3 4
C11 11 12 0.84p
C12 21 22 0.84p
C13 31 32 0.84p
C14 41 42 0.84p
C21 12 42 0.02p
C22 22 32 0.02p
C41 12 22 0.95p
C42 32 42 0.95p
K11 L21 L22 0.999999999
L11 11 12 3.1n
L12 21 22 3.1n
L13 31 32 3.1n
L14 41 42 3.1n
L21 12 42 200n
L22 22 32 200n
R11 11 12 14
R12 21 22 14
R13 31 32 14
R14 41 42 14
R21 12 42 880
R22 22 32 880
R31 1 11 0.065
R32 2 21 0.065
R33 3 31 0.065
R34 4 41 0.065
R41 12 22 1G
R42 32 42 1G
R51 1 4 1G
R52 2 3 1G
.ENDS ACM2012-900-2P
*
*-----
.END

```

図5 USB2.0 差動伝送回路のネットリスト

インピーダンスを有しますので、これを RIN1 と RIN2 で表しています。また、V1 と V2 で発生する逆位相のパルスに時間差を与えようとスキュー（差動信号間の時間的なずれ）が発生し、差動信号に共通モード（同相）成分が付加されます。共通モード成分は輻射ノイズの原因となるため、その発生を共通モードフィルタなどで極力抑える必要があります。しかし、適切でない共通モードフィルタを用いると信号波形にひずみが生じて伝送エラーの原因となります。ここではスキューによって発生した共通モード成分が共通モードフィルタによって除去される様子と、適切でないフィルタを使用した場合に発生する信号波形のひずみを、SPICE の過渡解析機能を使ってシミュレーションします。

USB2.0 の High Speed モードでは、電圧振幅が 0.4Vp-p の差動信号を用いて 480Mbps の速度でデータが伝送されます。従って、1bit あたりの時間幅はおおよそ 2.08nsec です。また、パルスの立ち上がり、立ち下り時間を 0.4nsec と仮定します。さらに、共通モード成分を付加するため V2 に 0.1nsec の遅延時間を設定しスキューを発生させます。したがって、信号源（パルス電源 + 出力インピーダンス）からの出力波形、すなわちノード 12 と 22 の電圧波形が図 4 に示す波形となるように V1 と V2 の条件を設定します。


また、USB2.0 規格ではディファレンシャルモードの特性インピーダンスが 90 と規定されています。これを模擬するため、2本のシングルエンドラインの特性インピーダンスをそれぞれ 45 とします。そして、信号源の出力インピーダンス、伝送路の特性インピーダンス、負荷抵抗をすべて 45 に設定します。

以上の条件を加味しますと、図 3 の回路をシミュレーションするためのネットリストは図 5 のように記述されます。図 5 の例では前半部分がメイン回路と解析コマンドで後半部分が共通モードフィルタのモデルとなっています。電源、伝送線路、抵抗は SPICE の基本素子を、また共通モードフィルタはライブラリに収録されている「ACM2012-900-2P」のモデルを用いました。図 3 の回路図より、共通モードフィルタへの差動信号入力はノード 12 と 22、また差動信号出力はノード 13 と 23 ですので、サブサーキットを呼び出すために前述のように「XCMF 12 22 23 13 ACM2012-900-2P」と記述しています。伝送線路の電気長（遅延時間）は 2nsec としました。過渡解析を 0.05nsec ステップで 10nsec まで行い、ノード 14 および 24 の電圧波形を出力するよう設定しました。


ところで、図 5 のネットリストでは V1 と V2 で発生するパルスの電圧振幅が 0.8Vp-p となるように記述されています。これは、信号源（パルス電源 + 出力インピーダンス）を負荷抵抗のみで終端した場合の出力電圧波形が、図 4 の波形となるようにするためであり、パルス電源の出力電圧が出力インピーダンスと負荷抵抗で

分圧されることを予め考慮した結果です。


図 5 のネットリストを SPICE で解析した結果を図 6 に示します。なお比較のために、ACM2012-900-2P（共通モードインピーダンス：90 @100MHz、USB 用途に適した製品）を用いた場合以外に、他の共通モードフィルタ（共通モードインピーダンス：600 @100MHz）を用いた場合と、共通モードフィルタなしの場合につ



(a) ACM2012-900-2P を用いた場合



(b) 他の共通モードフィルタを用いた場合



(c) コモンモードフィルタなしの場合

図 6 伝送波形のシミュレーション結果




図7 コモンモード成分の比較

いてシミュレーションした結果も示します。

3つのケース共に、伝送線路の遅延時間(2nsec)に相当する遅れの後、入力パルスに対応した出力波形が得られています。コモンモードフィルタなしの場合(図6(c))は、信号源で発生したスキューがそのまま受信端に現れていますが、コモンモードフィルタを用いる(図6(a),(b))ことでスキューが大幅に改善されます。また、USB用途に適したACM2012-900-2Pを用いた場合(図6(a))には目立った波形ひずみも見られず良好な伝送波形が得られますが、他のコモンモードフィルタを用いた場合(図6(b))では波形のなまりやひずみが発生しており伝送エラーを生じやすいことがわかります。したがって、システムに適した製品を選択することが重要であることがわかります。

次に、コモンモード成分の波形を図7に示します。コモンモード成分は $(V(14)+V(24))/2$ という計算式より得られます。図7では、ACM2012-900-2Pを用いた場合の他、コモンモードフィルタがない場合の結果も示しています。これより、100mVp-pであったコモンモード成分が、ACM2012-900-2Pを用いることで14mVp-pにまで抑制できることがわかります。したがって、ケーブルなどからの輻射ノイズを抑止する効果が期待されます。

● まとめ

本アプリケーションノートでは、「TDK SPICE Netlist Library」の概要、SPICEシミュレータでの基本的な使い方、および簡単な応用例について述べました。ここで述べた内容は基本的な事項でしたが、シミュレータの種類によってはライブラリをより便利にかつ高機能に使うことができます。例えば、外部ファイルのインクルード機能を使って簡単にモデルを読み込んだり、モデルを回路図エディタに登録してGUI上でシミュレーション回路を作成するという使い方です。また、ドライバやレシーバに実際のICのモデルを

用いたり、結合や損失を考慮した伝送路を使うことにより、より実際に即した回路シミュレーションが行えるようになります。詳しくはお使いのシミュレータのマニュアルなどをご参照ください。

<< 参考文献 >>

[1] <http://bwrc.eecs.berkeley.edu/Classes/lcBook/SPICE>

<< 技術支援ツールのホームページ >>

<http://www.tdk.co.jp/tst>



ご注意

- ◆ 本アプリケーションノートに記載されているデータやシミュレーション結果、および「TDK SPICE Netlist Library」に収録されているデータは、製品の特性を保証するものではありません。
- ◆ 本アプリケーションノートに記載されている内容や「TDK SPICE Netlist Library」を使用することにより生じる損害につきまして、TDKは一切の責任を負いません。
- ◆ 本アプリケーションノートに記載の内容は、2006年7月現在のものです。
- ◆ 本アプリケーションノートおよび「TDK SPICE Netlist Library」に記載の内容は、今後改良等のために予告なく変更することがあります。